




БИБЛИОТЕКА РАЗРАБОТЧИКА

Е. Ю. Хрусталева

ЯЗЫК ЗАПРОСОВ «1С:ПРЕДПРИЯТИЯ 8»



+CD

1C
ПАБЛИШИНГ

Е.Ю. Хрусталева

Язык запросов «1С:Предприятия 8»

Электронная книга в формате pdf; ISBN 978-5-9677-1992-9.

Электронный аналог печатного издания

«Язык запросов «1С:Предприятия 8»

(ISBN 978-5-9677-1987-5, М.: ООО «1С-Паблишинг», 2013;

артикул печатной книги по прайс-листу фирмы «1С»: 4601546108029;

по вопросам приобретения печатных изданий издательства «1С-Паблишинг» обращайтесь к партнеру

«1С», обслуживающему вашу организацию, или к другим партнерам фирмы «1С», в магазины

«1С Интерес», а также в книжные и интернет-магазины).

Запросы – это один из базовых механизмов «1С:Предприятия» наряду со встроенным языком, который позволяет читать и обрабатывать данные, хранящиеся в базе. Для составления запросов «1С:Предприятие» использует собственный язык, основанный на SQL.

Эта книга поможет начинающим разработчикам, не знакомым с SQL, освоить язык запросов «1С:Предприятия». Книга также будет полезна и тем, кто имеет опыт составления SQL-запросов в других средах разработки: язык запросов «1С:Предприятия» содержит значительное количество расширений, ориентированных на специфику финансово-экономических задач.

В книге рассматривается значительное количество практических примеров. Для создания примеров использована версия 8.3.3.687 платформы «1С:Предприятие».

Книга выпущена под редакцией Максима Радченко.

Дополнительные материалы

Приложение к книге включает демонстрационные конфигурации, иллюстрирующие примеры книги, и учебную версию платформы "1С:Предприятие 8.3".

Скачайте материалы и учебную версию на странице http://its.1c.ru/book_demo/, раскройте архив и следуйте инструкциям по установке.

Интернет-конференция для начинающих разработчиков
<http://devtrainingforum.v8.1c.ru/forum>.

Оглавление

Глава 1. Механизм запросов	7
Как хранятся данные в «1С:Предприятии»	7
Исходные таблицы для запросов	17
Реальные таблицы	19
Виртуальные таблицы	20
Язык запросов «1С:Предприятия»	21
Общая схема выполнения запросов	22
Синтаксис текста запросов	24
Примеры использования языка запросов для получения данных из одной таблицы	27
Как получить все данные из таблицы	27
Как получить только определенные поля для всех записей из таблицы	29
Как расположить полученные записи в нужном порядке	33
Как упорядочить записи таблицы по ссылочному полю	36
Как получить текстовое представление ссылочного поля	38
Как получить только первые несколько записей с наибольшими значениями некоторого поля	41
Как получить записи, в которых определенные поля не содержат одинаковых значений	42
Как получить общее количество записей в таблице и количество записей с различным значением некоторого поля	44
Как получить записи из таблицы, отобранные по некоторому условию	45
Как получить записи таблицы, содержащие строки, соответствующие заданному шаблону	47
Как задать произвольное значение отбора записей из таблицы	49
Как получить данные из табличной части некоторого документа	52
Как получить данные из табличной части документа в качестве вложенной таблицы	55
Как получить записи иерархической таблицы и расположить их в порядке иерархии	57
Как отобрать записи иерархической таблицы по условию	62
Как узнать среднюю цену, по которой продавался товар	68

Как узнать общее количество и сумму продаж каждого товара в разрезе покупателей.....	71
Как узнать среднюю цену поступления товара, не группируя сами записи.....	74
Расчет итогов для иерархического справочника	76
Расчет итогов по нескольким полям	78
Расчет общих итогов	80
Примеры использования выражений в списке полей выборки запроса.....	82
Примеры использования языка запросов	
для получения данных из нескольких таблиц.....	87
Как использовать данные одного запроса внутри другого запроса	87
Как получить данные из разных таблиц для одного и того же поля	91
Как получить данные из разных таблиц, связанных несколькими соединениями.....	100
Как получить данные из таблицы, на которую ссылается поле другой таблицы	110
Как получить данные из разных таблиц, не связывая, а дополняя их.....	114
Временные таблицы и пакетные запросы.....	117
Глава 2. Работа с запросами во встроенном языке	125
Конструктор запроса	125
Создание простого запроса.....	127
Связи источников запроса	132
Объединение запросов.....	134
Создание пакетного запроса, использующего временную таблицу	140
Выполнение запросов из встроенного языка	146
Создание запроса.....	147
Передача параметров в запрос	148
Получение выборки из результата запроса	151
Обход выборки из результата запроса.....	152
Линейный (прямой) порядок обхода	155
Иерархический порядок обхода	156
Обход по группировкам	160
Обход выборки результата запроса, содержащего данные табличной части	162
Обработка результатов запроса с помощью конструктора запроса	164
Обход выборки.....	164
Вывод в табличный документ	167
Вывод в диаграмму	172
Выгрузка результата запроса в таблицу или дерево значений	176
Отладка запросов	179
Использование временных таблиц с помощью встроенного языка	180
Использование таблицы значений в качестве источника временной таблицы	184
Примеры решения различных задач с использованием запросов	189
Поиск всех родителей для элемента иерархического справочника.....	189
Создание запроса из произвольного источника.....	192
Создание кросс-отчета.....	196
Вывод итогов по периодам с заданной периодичностью.....	201

Глава 3. Решение прикладных задач	207
Хранение информации.....	208
Регистры сведений.....	208
Получение данных из независимых неперiodических регистров сведений	209
Получение данных из периодических регистров сведений	212
Получение данных из регистров сведений, подчиненных регистратору	219
Планы видов характеристик	221
Пример 1	222
Пример 2	223
Пример 3	225
Получение значений характеристик из регистра сведений	226
Учет движения средств	229
Регистры накопления	229
Получение движений регистра накопления	231
Получение остатков.....	234
Получение оборотов.....	246
Получение остатков и оборотов	255
Бухгалтерский учет.....	265
Планы счетов.....	265
Планы видов характеристик – виды субконто	270
Регистры бухгалтерии.....	273
Получение движений регистра бухгалтерии	276
Получение остатков.....	277
Получение оборотов.....	291
Получение оборотов между корреспондирующими счетами.....	298
Получение остатков и оборотов	305
Получение движений с субконто	313
Сложные периодические расчеты	316
Планы видов расчета.....	316
Регистры расчета	322
Получение данных из регистра расчета	324
Получение данных о фактическом периоде действия записи для расчета	327
Получение данных графика для расчета записи	330
Получение базы для расчета записей	331
Перерасчеты	337
Глава 4. Оптимизация запросов	339
Индексирование таблиц.....	340
Способы индексирования таблиц	344
Эффективное использование индексов	348
Пример 1	349
Пример 2	350
Пример 3	351
Пример 4	352

Пример 5	353
Общие рекомендации.....	354
Причины неоптимальной работы запросов и основные направления их оптимизации.....	355
Общие рекомендации	355
Не использовать запросы в цикле	356
Не использовать в запросе функции от параметров	358
Использовать параметры виртуальных таблиц	359
Соответствие индексов и условий запроса.....	360
Не использовать соединения с вложенными запросами и с виртуальными таблицами ..	360
Соединения с вложенными запросами.....	360
Соединения с виртуальными таблицами	363
Не использовать вложенные запросы в условиях соединения	365
Исключить получение поля «Ссылка» через точку	365
Ограничить получение данных через точку от полей составного ссылочного типа	366
Исключить вывод ссылочных полей в отчет.....	368

Глава 1.

Механизм запросов

Как хранятся данные в «1С:Предприятии»

Прикладные решения, разработанные на платформе «1С:Предприятие», работают с данными, которые интерактивно вводит пользователь, заполняя различные формы ввода справочников, документов и т. д. Из этих форм данные с помощью встроенного языка записываются в базу данных и хранятся в ней. На основе введенных данных пользователю обычно требуется получить некоторую обобщенную информацию, необходимые ему итоговые данные, отчеты и т. д. Для этого используется механизм запросов, который рассматривается в данной книге.

Поэтому прежде чем начинать осваивать язык запросов, важно понять, как хранятся данные в «1С:Предприятии». Для хранения данных «1С:Предприятие» использует реляционные базы данных. *Реляционная база данных* представляет собой совокупность различной информации, представленной в виде двумерных *таблиц*. Таблица базы данных состоит из набора строк и столбцов. Каждая строка (*запись*) этой таблицы характеризуется рядом значений, содержащихся в ее столбцах (*полях*).

Для примера можно привести данные о клиентах компании, хранящиеся в справочнике клиентов. В самом упрощенном виде данные о клиентах хранятся в одной – *основной* таблице, где каждому клиенту соответствует одна запись, имеющая один и тот же набор полей, например, Код, Наименование, Адрес, Телефон и т. п. (табл. 1.1).

Таблица 1.1. Справочник клиентов в информационной базе (основная таблица)

Ссылка	Код	Наименование	Адрес	Телефон
Ref1	000000001	Соколов Иван Андреевич	Москва, ...	8-916-222-33-55
Ref2	000000002	Орлов Сергей Иванович	Москва, ...	8-926-555-66-77
Ref3	000000003	Маслова Ирина Николаевна	Санкт-Петербург, ...	8-915-4447799

Рассмотрим подробнее, как формируется структура таблицы справочника в информационной базе «1С:Предприятия».

Например, при создании в конфигураторе справочника Клиенты платформа «1С:Предприятие» автоматически создает в информационной базе основную таблицу этого справочника с полями Ссылка, Код, Наименование, ПометкаУдаления, Предопределенный и ВерсияДанных.

ПРИМЕЧАНИЕ

Имена полей Код, Наименование и т. п. мы используем для простоты изложения. На самом деле имена полей таблиц в базе данных будут техногенными, например, _Code, _Description и т. п.

Затем пользователь в режиме 1С:Предприятие заполняет справочник данными, которые сохраняются в базе данных (рис. 1.1).




Рис. 1.1. Структура простого справочника в конфигураторе, в «1С:Предприятии» и в информационной базе

Поле Ссылка является уникальным идентификатором записи о клиенте, поля Код, Наименование, ПометкаУдаления, Предопределенный и ВерсияДанных являются стандартными реквизитами, которые платформа добавляет в любой справочник.

На рис. 1.1 также показаны три понятия, которые важно отличать. В конфигураторе создается *объект конфигурации* Справочник Клиенты. Данные этого объекта конфигурации (все записи справочника) вводятся в режиме 1С:Предприятие, затем эти данные записываются в таблицу справочника Клиенты в информационной базе. В этой таблице *объектом базы данных* является одна запись (данные одного элемента справочника), однозначно идентифицирующаяся значением поля Ссылка.

Каждый элемент справочника, как правило, содержит некоторую дополнительную информацию, которая подробнее описывает этот элемент. Например, все элементы справочника Клиенты могут содержать дополнительную информацию об адресе и телефоне каждого клиента. Для описания этой информации используются *реквизиты* справочника.

При добавлении реквизитов справочника в конфигураторе платформа создает поля соответствующего типа в основной таблице справочника (рис. 1.2).




Рис. 1.2. Структура справочника, имеющего реквизиты, в конфигураторе, в «1С:Предприятии» и в информационной базе

Если справочник является иерархическим, то платформа добавляет в структуру основной таблицы справочника в информационной базе поле Родитель и поле ЭтоГруппа (в случае, если справочник имеет тип иерархии Иерархия групп и элементов), рис. 1.3.




Рис. 1.3. Структура иерархического справочника в конфигураторе и в информационной базе

Например, справочник Товары является иерархическим с иерархией групп и элементов. При этом в таблице базы данных для записей, являющихся группой, поле ЭтоГруппа принимает значение Истина; для записей, не являющихся группой, поле ЭтоГруппа принимает значение Ложь, а поле Родитель является ссылкой на родительскую запись. Благодаря этому можно получить информацию о дочерних записях и родителях для каждого элемента справочника (рис. 1.4).

ПРИМЕЧАНИЕ

Поле Родитель на самом деле хранит ссылку на родительскую запись, но для большей ясности в таблице на рис. 1.4 в этом поле отражено представление ссылки в виде наименования.

Если справочник является подчиненным, например справочник РасчетныеСчета подчинен справочнику Поставщики, то в основную таблицу подчиненного справочника платформа добавляет поле Владелец, которое ссылается на элемент справочника-владельца (рис. 1.5).

1С:Предприятие

Товары

Создать [иконки] Найти... Все действия ?

Наименование	Код	Производитель
Товары		
Обувь	000000001	
Детская обувь	000000009	
Пинетки	000000010	
Кроссовки	000000004	
Сапоги	000000003	Италия
Туфли	000000002	Германия
Продукты	000000005	
Масло	000000006	Россия
Молоко	000000007	Россия
Сметана	000000008	

База данных

Иерархический справочник "Товары" (с иерархией групп и элементов)

Ссылка	Код	Наименование	...	Родитель	ЭтоГруппа	Производитель
Ref1	000000001	Обувь			ИСТИНА	
Ref9	000000009	Детская обувь		Обувь	ИСТИНА	
Ref10	000000010	Пинетки		Детская обувь	ЛОЖЬ	
Ref4	000000004	Кроссовки		Обувь	ЛОЖЬ	
Ref3	000000003	Сапоги		Обувь	ЛОЖЬ	Италия
Ref2	000000002	Туфли		Обувь	ЛОЖЬ	Германия
Ref5	000000005	Продукты			ИСТИНА	
Ref2	000000006	Масло		Продукты	ЛОЖЬ	Россия
Ref3	000000007	Молоко		Продукты	ЛОЖЬ	Россия
Ref4	000000008	Сметана		Продукты	ЛОЖЬ	

Рис. 1.4. Содержимое иерархического справочника
в «1С:Предприятии» и в информационной базе

Благодаря этому можно получить информацию, какие элементы справочника-владельца владеют какими элементами подчиненного справочника, например, какие расчетные счета относятся к конкретному поставщику (рис. 1.6).




Рис. 1.5. Структура подчиненного справочника в конфигураторе и в информационной базе



Рис. 1.6. Содержание подчиненного справочника и справочника-владельца в «1С:Предприятии» и в информационной базе

ПРИМЕЧАНИЕ

Поле Владелец подчиненного справочника на самом деле хранит ссылку на запись справочника-владельца, но для большей ясности в таблице на рис. 1.6 в этом поле отражено представление ссылки в виде наименования.

Кроме реквизитов каждый элемент справочника может содержать некоторый набор информации, которая одинакова по своей структуре, но различна по количеству и относится к разным элементам справочника. Например, каждый элемент справочника Поставщики может содержать информацию о договорах, заключенных с этим поставщиком. Для каждого поставщика состав информации и количество записей в ней будет разным, а структура информации (например, дата начала и окончания действия договоров) – одинакова. Для описания подобной информации могут быть использованы *табличные части* справочника.

При добавлении в справочник табличной части в информационной базе создается подчиненная таблица со стандартными полями Ссылка и НомерСтроки и реквизитами табличной части, заданными в конфигураторе. Таблица, содержащая табличную часть, связана по полю Ссылка с основной таблицей. Благодаря этому можно получить информацию из табличной части, относящуюся к конкретному элементу справочника (рис. 1.7).

ПРИМЕЧАНИЕ

В поле Ссылка подчиненной таблицы, содержащей табличную часть, на самом деле хранится ссылка на запись основной таблицы, но для большей ясности в таблице на рис. 1.7 в этом поле отражено представление ссылки в виде наименования.

В информационной базе создается столько подчиненных таблиц, сколько табличных частей задано у справочника.

Таким образом, на примере справочника, имеющего табличную часть, мы видим, что одному объекту конфигурации в информационной базе могут соответствовать несколько таблиц – *основная* и одна или несколько *подчиненных* ей по полю Ссылка таблиц. При этом одному объекту базы данных соответствует одна запись в основной таблице и одна или несколько записей в подчиненных таблицах, содержащих табличные части (см. рис. 1.7).




Рис. 1.7. Справочник с табличной частью в конфигураторе, в «1С:Предприятия» и в информационной базе

Теперь рассмотрим другой пример, когда *поле ссылочного типа* служит для связи данных двух разных объектов конфигурации. Важно понимать, что ссылочные типы данных не существуют изначально в конфигурации, а появляются при создании соответствующих объектов конфигурации. Причем для каждого объекта конфигурации во встроенном языке создается свой тип ссылки. То есть при создании справочника Товары появляется ссылочный тип данных СправочникСсылка.Товары, при создании справочника Клиенты – тип СправочникСсылка.Клиенты, при создании документа Событие – тип ДокументСсылка.Событие и т.д. Поля, содержащие данные такого типа, мы будем для краткости называть иногда *ссылочными полями*.

Например, в конфигурации существует документ, имеющий поле, ссылающееся на справочник Клиенты (рис. 1.8).




Рис. 1.8. Документ, имеющий поле ссылочного типа, в конфигураторе, в «1С:Предприятии» и в информационной базе

ПРИМЕЧАНИЕ

Поле Клиент документа Событие на самом деле хранит ссылку на запись справочника Клиенты, но для большей ясности в таблице на рис. 1.8 в этом поле отражено представление ссылки в виде наименования.

В приведенном примере поле Клиент документа Событие имеет ссылочный тип СправочникСсылка.Клиенты, а значениями этого поля являются ссылки на конкретные элементы справочника Клиенты. В этом случае, обращаясь к полю Клиент в документе, мы можем получить любые данные о клиенте, на которого ссылается данное поле.

Таким образом, мы рассмотрели примеры хранения в информационной базе «1С:Предприятия» ссылочных типов данных, таких как справочники, документы, планы видов характеристик и т. д.

В заключение рассмотрим, как хранятся в информационной базе «1С:Предприятия» нессылочные данные, доступ к которым нельзя получить через поле Ссылка.

Например, периодический регистр сведений предназначен для хранения данных (ресурсов) в разрезе измерений с привязкой ко времени. Благодаря стандартному полю Период, регистр сведений может хранить не только актуальные значения данных, но и историю их изменения во времени.




Рис. 1.9. Периодический регистр сведений в конфигураторе, в «1С:Предприятии» и в информационной базе

На основе объекта конфигурации Регистр сведений платформа создает в информационной базе таблицу, в которой может храниться произвольная информация, «привязанная» к набору измерений и периоду. Например, периодический регистр сведений Цены, имеющий измерение Товар и ресурс Цена, хранит изменяющуюся во времени информацию о ценах на товары (рис. 1.9).

ПРИМЕЧАНИЕ

Поле Товар в таблице регистра на самом деле хранит ссылку на запись справочника Товары, но для большей ясности на рис. 1.9 в этом поле отражено представление ссылки в виде наименования.

Помимо измерений, ресурсов и реквизитов, заданных в конфигураторе, у таблицы периодического регистра сведений в информационной базе создается стандартное поле Период, благодаря этому регистр может хранить одинаковую информацию для одних и тех же измерений, но для различных периодов.

Как мы видим, в таблице регистра нет поля Ссылка, посредством которого мы можем сослаться на конкретную запись регистра. Ключом записи, однозначно идентифицирующим запись, является в случае периодического регистра сведений совокупность значений измерений регистра и периода.

В целом для нессылочных данных мы не можем получить какую-то конкретную запись из таблицы. Но мы можем получить некоторый набор записей по какому-либо условию (например, отобразить данные регистра сведений по периоду) и затем перебирать его в цикле.

Исходные таблицы для запросов

Прежде чем переходить к конкретным примерам использования языка запросов, остановимся также на составе таблиц базы данных, являющихся источниками запросов. Состав таблиц, доступных для запроса, и их описание мы можем увидеть в синтакс-помощнике в разделе Работа с запросами ► Таблицы запросов.

Важно понимать, что прямого доступа к физическим таблицам, в которых хранится информация в базе данных, из «1С:Предприятия» получить нельзя. Это связано с тем, что в прикладном решении могут использоваться разные СУБД, имеющие свою специфику, а текст запроса должен быть универсальным и одинаково работать на любой используемой СУБД. Поэтому при выполнении запроса платформа автоматически транслирует текст запроса в набор инструкций, которые «понимает» конкретная СУБД. Кроме того,

физические таблицы и поля в них имеют техногенные имена, из которых непонятно, что именно хранится в данном поле.

Поэтому с помощью запросов мы обращаемся к данным не напрямую, а через специальную «прослойку» в виде *таблиц языка запросов*. Этот процесс можно представить на следующей схеме (рис. 1.10).




Рис. 1.10. Доступ к данным в «1С:Предприятии»

Таким образом, все таблицы, к которым можно обратиться с помощью языка запросов, являются придуманными, воображаемыми, в большей или меньшей степени соответствующими реальным физическим таблицам СУБД. Однако по степени похожести на физические таблицы их принято разделять на *реальные* и *виртуальные* таблицы.

Реальные таблицы

Отличительной особенностью реальных таблиц является то, что они содержат данные какой-либо одной физической таблицы, хранящейся в базе данных, и то, что реальная таблица очень похожа на свою физическую таблицу. Например, реальной является таблица `Справочник.Клиенты`, соответствующая справочнику `Клиенты`, или таблица `РегистрСведений.Цены`, соответствующая регистру сведений `Цены`.

Для примера сравним структуру реальной и физической таблицы, хранящей данные справочника (табл. 1.2).

Таблица 1.2. Реальная и физическая таблица справочника «Клиенты»

Реальная таблица	Физическая таблица
Справочник.Клиенты	Reference<n>
Ссылка	Id<suff>
ВерсияДанных	_Version
ПометкаУдаления	_Marked
Предопределенный	_IsMetadata
Родитель	_ParentId<suff>
Владелец	_OwnerId<suff>
ЭтоГруппа	_Folder
Код	_Code
Наименование	_Description
<Имя реквизита>	_Fld<n><suff>
<Имя общего реквизита>	_Fld<n><suff>
Представление	–
<Имя табличной части>	–

Как мы видим, не все поля реальной и физической таблицы соответствуют друг другу. Например, поле `Представление` – виртуальное, то есть оно не хранится в физической таблице базы данных, а генерируется в момент выполнения запроса.

ПОДРОБНЕЕ

О поле `Представление` рассказано в разделе «Как получить текстовое представление ссылочного поля», стр. 38.

О поле `Имя табличной части` рассказано в разделе «Как получить данные из табличной части документа в качестве вложенной таблицы», стр. 55.

Но в целом реальная таблица очень похожа на физическую по набору полей, а также количество записей в обеих таблицах одинаково.

Реальные таблицы подразделяются на *объектные* (ссылочные) и *необъектные* (нессылочные).

В объектных (ссылочных) таблицах представлена информация ссылочных типов данных (справочники, документы, планы видов характеристик и т. д.). А в необъектных (нессылочных) – всех остальных типов данных (константы, регистры и т. д.).

Отличительной особенностью объектных (ссылочных) таблиц является то, что они включают в себя стандартное поле Ссылка, которое позволяет однозначно идентифицировать каждую запись (данные об объекте базы данных). Эти таблицы могут быть иерархическими, подчиненными, и поля таких таблиц могут содержать вложенные таблицы (табличные части).

Виртуальные таблицы

Виртуальные таблицы формируются в момент выполнения запроса на основе реальных таблиц базы данных. Например, виртуальная таблица РегистрСведений.Цены.СрезПоследних формируется на основе таблицы регистра сведений Цены, рис. 1.11.

ПРИМЕЧАНИЕ

Поле Товар в таблице регистра на самом деле хранит ссылку на запись справочника Товары, но для большей ясности на рис. 1.11 в этом поле отражено представление ссылки в виде наименования.

Как мы видим из рисунка 1.11, при заполнении цен товаров в регистре сведений Цены в «1С:Предприятии» данные за период по каждому товару сохраняются в физической таблице регистра сведений в базе данных.

Мы уже рассказывали выше про периодический регистр сведений (см. рис. 1.9). Благодаря стандартному полю Период регистр может хранить изменяющуюся во времени информацию для одних и тех же измерений, но для различных периодов.

При обращении запросом к виртуальной таблице РегистрСведений.Цены.СрезПоследних мы получим не все записи физической таблицы, а только последние по времени данные о ценах на товары. То есть срез последних записей регистра сведений возвращает по каждому значению измерения (Товары) одну наиболее позднюю (по времени, по значению поля Период) запись.

Таким образом, виртуальные таблицы мало похожи на какую-то физическую таблицу и содержат совсем иной состав записей, чем реальные таблицы.

образом и пр. Далее полученные данные могут быть проанализированы для решения различных прикладных задач, построения отчетов и т. п.

Однако следует иметь в виду, что с помощью запросов можно только прочитать нужную информацию из базы данных, но изменить ее и записать обратно при помощи запроса нельзя – для этого нужно использовать средства встроенного языка.

Общая схема выполнения запросов

Запрос формируется и выполняется разработчиком из встроенного языка. Для этого предназначены следующие программные объекты:

- Запрос,
- РезультатЗапроса,
- ВыборкаИзРезультатаЗапроса.

Не углубляясь в детали, рассмотрим самую распространенную и простейшую схему выполнения запроса.

ПОДРОБНЕЕ

Другие варианты выполнения запросов и обработки их результатов будут рассмотрены позднее в разделе «Выполнение запросов из встроенного языка», стр. 146.

1. Сначала во встроенном языке создается объект Запрос (листинг 1.1).

Листинг 1.1. Создание запроса

```
Запрос = Новый Запрос;
```

2. У объекта Запрос есть свойство Текст, в которое нужно поместить текст запроса, написанный на языке запросов. В тексте запроса описывается, какие данные, из каких таблиц нужно получить и как эти данные представить (листинг 1.2).

Листинг 1.2. Заполнение текста запроса

```
Запрос.Текст =  
    "ВЫБРАТЬ  
    |     Наименование  
    |ИЗ  
    |     Справочник.Товары";
```

3. Далее запрос выполняется с помощью метода Выполнить() объекта Запрос. Именно в этот момент и происходит чтение данных из базы данных. Прочитанные данные возвращаются в виде объекта РезультатЗапроса, содержащего выбранные данные из базы данных (листинг 1.3).

Листинг 1.3. Выполнение запроса

```
РезультатЗапроса = Запрос.Выполнить();
```

4. Чтобы обработать данные, содержащиеся в объекте РезультатЗапроса, из результата запроса получается выборка с помощью метода Выбрать(), который возвращает новый объект ВыборкаИзРезультатаЗапроса, то есть коллекцию данных, предназначенную для последовательного обхода ее элементов (листинг 1.4).

Листинг 1.4. Получение выборки из результата запроса

```
Выборка = РезультатЗапроса.Выбрать();
```

5. Далее выборка обходится с помощью цикла Пока Выборка.Следующий() Цикл, а в теле цикла производятся какие-то действия над данными, полученными с помощью запроса (листинг 1.5).

Листинг 1.5. Обход выборки

```
Пока ВыборкаЗапроса.Следующий() Цикл
    Сообщение = Новый СообщениеПользователю;
    Сообщение.Текст = ВыборкаЗапроса.Наименование;
    Сообщение.Сообщить();
КонецЦикла;
```

В результате, если соединить вместе текст листингов 1.1–1.5, мы получим процедуру встроенного языка, в которой создается и выполняется запрос, в данном примере выводящий наименование всех товаров из справочника Товары в окно сообщений (листинг 1.6).

Листинг 1.6. Вывод наименований всех товаров в окно сообщений

```
Запрос = Новый Запрос;
Запрос.Текст =
    "ВЫБРАТЬ
     | Наименование
     | ИЗ
     | Справочник.Товары";

РезультатЗапроса = Запрос.Выполнить();
ВыборкаЗапроса = РезультатЗапроса.Выбрать();

Пока ВыборкаЗапроса.Следующий() Цикл
    Сообщение = Новый СообщениеПользователю;
    Сообщение.Текст = ВыборкаЗапроса.Наименование;
    Сообщение.Сообщить();
КонецЦикла;
```

Пример этой процедуры находится в демонстрационной конфигурации «Язык запросов», прилагающейся к книге, в обработке Работа с запросами. Результат выполнения данной процедуры представлен на рис. 1.12.




Рис. 1.12. Вывод наименований всех товаров в окно сообщений

Чтобы не усложнять восприятие материала, пока мы не будем подробно останавливаться на выводе и обработке результатов запросов. Сначала мы начнем изучать сам язык запросов, то есть научимся правильно составлять текст запросов. Для выполнения запросов и просмотра результатов мы будем использовать специальную обработку Консоль запросов. Данная обработка помогает отлаживать и просматривать результаты выполнения запросов в режиме 1С:Предприятие.

Обработка Консоль запросов добавлена в демонстрационную конфигурацию «Язык запросов», прилагающуюся к книге на компакт-диске. Самая последняя версия этой обработки опубликована на ИТС (<http://its.1c.ru/db/metod81#content:4500:1>).

Синтаксис текста запросов

Язык запросов «1С:Предприятия» основан на стандартном SQL, но при этом содержит значительное количество расширений, ориентированных на финансово-экономические задачи, и значительно облегчает разработку бизнес-приложений.

Из определения следует, что язык запросов – мощный инструмент, предоставляющий разнообразные возможности получения данных. Но мы начнем его изучение с самых простых примеров, основанных на реальных небольших задачах, расположенных от простого к сложному. Но сначала немного теории.

Текст запроса состоит из нескольких частей (секций):

- описание запроса,
- объединение запросов,
- упорядочивание результатов,
- автоупорядочивание,
- описание итогов.

Обязательной частью запроса является только первая – описание запроса. Все остальные присутствуют в запросе по необходимости. Назначение каждой секции запроса будет рассмотрено ниже на конкретных примерах.

Для ознакомления приведем запрос, в котором присутствуют все указанные секции (рис. 1.13).

Секции запроса	

ВЫБРАТЬ	
Приход.Ссылка КАК Документ,	
Приход.Товар КАК Товар,	
Сумма(Приход.Количество) КАК КоличествоВсего,	
Сумма(Приход.Сумма) КАК СуммаВсего	
ИЗ	Описание запроса
Документ.Приходная.Накладная.Состав КАК Приход	
СГРУППИРОВАТЬ ПО	
Приход.Ссылка,	
Приход.Товар	

ОБЪЕДИНИТЬ ВСЕ	Объединение запросов

ВЫБРАТЬ	
Расход.Ссылка,	
Расход.Товар,	
Сумма(Расход.Количество),	
Сумма(Расход.Сумма)	
ИЗ	Описание запроса
Документ.Расходная.Накладная.Состав КАК Расход	
СГРУППИРОВАТЬ ПО	
Расход.Ссылка,	
Расход.Товар	

УПОРЯДОЧИТЬ ПО	
Документ,	Упорядочивание результатов
Товар	

АВТОУПОРЯДОЧИВАНИЕ	Автоупорядочивание

ИТОГИ	
Сумма(КоличествоВсего),	
Сумма(СуммаВсего)	Описание итогов
ПО	
Документ	

Рис. 1.13. Секции запроса

Синтаксически текст запроса состоит из набора *секций*, имеющих определенное назначение, например, выбрать записи из базы данных, отсортировать их, подсчитать итоги и т. д. Секции состоят из *предложений*, которые, в свою очередь, содержат *ключевые слова* (например, ВЫБРАТЬ, ИЗ, ГДЕ и т. п.), обозначающие определенное действие, которое нужно выполнить с базой данных. Ключевое слово, с которого начинается предложение, обычно дает название предложению языка запросов.

ПОДРОБНЕЕ

Применение различных синтаксических конструкций языка запросов подробно описано во встроенной справке Справка ► Содержание справки ► 1С:Предприятие ► Встроенный язык ► Работа с запросами ► Синтаксис текста запросов.

Одной из существенных особенностей языка запросов «1С:Предприятия» является то, что все ключевые слова имеют два варианта написания: на русском и английском языках. Поэтому язык написания запроса – дело привычки и вкуса, а результат выполнения запроса будет одинаков в обоих случаях.

В книге мы будем использовать ключевые слова языка запросов на русском языке, а для тех, кто хочет использовать англоязычный вариант написания ключевых слов, они подробно описаны во встроенной справке.

ПОДРОБНЕЕ

Встроенная справка: Справка ► Содержание справки ► 1С:Предприятие ► Встроенный язык ► Работа с запросами ► Синтаксис текста запросов ► Двухязычное представление ключевых слов.

Итак, начнем изучать язык запросов на конкретных примерах, от самых простых к более сложным. В процессе изучения мы узнаем, как и для чего используются различные ключевые слова языка запросов.

Для составления примеров будем использовать демонстрационную конфигурацию «Язык запросов», прилагающуюся к книге на компакт-диске.

Примеры использования языка запросов для получения данных из одной таблицы

Как получить все данные из таблицы

В нашей демонстрационной конфигурации существует справочник Клиенты. Предположим, нам нужно получить все данные из таблицы базы данных, соответствующей этому справочнику. Это можно сделать с помощью простейшего запроса, а результат посмотреть в консоли запросов.

Обработка Консоль запросов добавлена в демонстрационную конфигурацию «Язык запросов», и ее можно вызвать из группы команд Сервис. Но для того чтобы полностью использовать все функциональные возможности консоли запросов, ее нужно запускать в режиме Толстый клиент.

Запустим демонстрационную конфигурацию в этом режиме. Для этого в диалоге запуска «1С:Предприятия» нужно создать новую информационную базу с нашей конфигурацией, добавив базу из шаблона, установленного с диска, и указать в качестве основного режима запуска информационной базы Толстый клиент (рис. 1.14).




Рис. 1.14. Основной режим запуска демонстрационной конфигурации – «Толстый клиент»

Вызовем консоль запросов из группы команд Сервис и в среднем окне Текст запроса введем следующий текст (листинг 1.7).

Листинг 1.7. Вывод всех данных из таблицы

```
ВЫБРАТЬ  
    Справочник.Клиенты.*
```

Нажмем кнопку **Выполнить** и в нижнем окне **Результат запроса** увидим результат выполнения нашего запроса (рис. 1.15).




Рис. 1.15. Вывод всех данных из таблицы

После выполнения запроса в заголовке окна **Результат запроса** появится также количество строк в результате запроса и время выполнения запроса.

Теперь рассмотрим подробно текст нашего запроса (см. листинг 1.7).

Текст любого запроса всегда содержит секцию *описания запроса*, в которой определяются источники данных для запроса, список полей выборки и т.д. Все секции запроса приведены на рис. 1.13, стр. 25.

Описание запроса начинается с ключевого слова **ВЫБРАТЬ**. За ним следует список имен полей выборки запроса, перечисленных через запятую. Таким образом, с помощью ключевого слова **ВЫБРАТЬ** определяются требуемые поля результата запроса.

Справочник.Клиенты – это имя одной из исходных таблиц запроса, описанных выше. В языке запросов имена таблиц формируются по принципу: <Имя класса объектов>.<Имя объекта конфигурации>. Полное имя поля содержит имя таблицы и имя поля. Например, в строке Справочник.Клиенты.Код Справочник.Клиенты – это имя таблицы, а Код – это имя поля.

ПОДРОБНЕЕ

Посмотреть состав таблиц, доступных для запроса, и их описание можно в синтаксис-помощнике в разделе **Работа с запросами** ► **Таблицы запросов**.

В данном случае нам нужны все поля таблицы, поэтому вместо перечисления имен полей можно использовать звездочку «*».

Обратите внимание на структуру написания запроса. Правила оформления запросов предписывают все ключевые слова выделять заглавными буквами, каждое поле из списка выборки начинать с новой строки, со сдвигом относительно слова **ВЫБРАТЬ**.

Если вы напишете в одну строку – «выбрать справочник.клиенты.*», то платформа вас поймет и так, и результат запроса не изменится (см. рис. 1.15). Но так писать – это моветон.

С остальными правилами оформления запросов мы будем знакомиться по ходу следующих примеров.

Итак, в данном примере мы получили все данные (стандартные и созданные разработчиком) из объектной таблицы, кроме виртуальных полей. Такие поля нельзя выбрать с помощью символа «*», их имена нужно указывать явно. Эта возможность рассматривается в следующем примере.

Как получить только определенные поля для всех записей из таблицы

В реальных задачах обычно требуется получить не все, а только некоторые конкретные поля из таблицы. В данном примере для всех записей справочника Клиенты получим только поля Наименование, Телефон и Представление.

Это можно сделать с помощью следующего запроса (листинг 1.8).

Листинг 1.8. Вывод определенных полей для всех записей из таблицы

ВЫБРАТЬ

Справочник.Клиенты.Наименование,
Справочник.Клиенты.Телефон,
Справочник.Клиенты.Представление

В данном запросе после ключевого слова **ВЫБРАТЬ** перечислены полные имена требуемых полей результата запроса.

Результат выполнения запроса будет выглядеть следующим образом (рис. 1.16).

Запрос: (Записей в результате: 3)

Наименование	Телефон	Представление
Сokolov Иван Андреевич	8-916-222-33-55	Сokolov Иван Андреевич
Орлов Сергей Иванович	8-926-555-66-77	Орлов Сергей Иванович
Маслова Ирина Николаевна	8-915-4447793	Маслова Ирина Николаевна

Рис. 1.16. Вывод определенных полей для всех записей из таблицы